



Le GTL par l'exemple

Initiation à la production de
graphiques « nouvelle génération »
avec SAS 9.1.3

Sommaire



1. Grands principes
2. Éléments graphiques
3. Éléments de texte
4. Juxtaposition d'éléments



1- Grands principes

© Olivier Decourt 2008 éditions Dunod
Comptabilité à Reporting avec SAP

1.1 - Stockage & utilisation (1/2)

```
PROC TEMPLATE ;  
  DEFINE statGraph  
    nomGraphique ;  
  ...  
  END ;  
RUN ;
```

1

sauvegarde




2

Template store
(SASUSER.TEMPLATE)

```
DATA _NULL_ ;  
  SET ... ;  
  FILE PRINT ODS=(TEMPLATE="nomGraphique" ...) ;  
  PUT _ODS_ ;  
RUN ;
```

3




1.1 - Stockage & utilisation (2/2)


- Le graphique est généré via un driver Java → esthétique de grande classe
- Le *template* (= modèle) peut être paramétré (cf. macro-programmes)
- Le GTL intègre des *layouts* (= motifs) pour agencer proprement plusieurs graphiques, avec titres, légendes & autres fioritures

1.2 - Les motifs


- 3 types de motifs...



LAYOUT OVERLAY
(éléments superposés)



LAYOUT LATTICE
(éléments juxtaposés,
espaces identiques)



LAYOUT GRIDDED
(éléments juxtaposés,
espaces adaptables)

1.3 - Les paramètres

- Paramétrer un template de graphique
 - Avec des variables internes
 - type texte
 - déclarées dans DYNAMIC
 - affectées au cours de l'étape Data
 - Avec des macro-variables
 - type texte
 - déclarées dans MVAR
 - affectées avant l'étape Data (MV système, %LET, CALL SYMPUTX, SELECT...INTO)

1.4 - Syntaxe (1/2)

```
PROC TEMPLATE ;  
  DEFINE STATGRAPH nomGraphique ;  
    < MVAR nomMV1 < nomMV2 < ... > > ; >  
    < DYNAMIC nomVar1 < nomVar2 < ... > > ; >  
    LAYOUT OVERLAY | GRIDDED | LATTICE  
      < / options > ;  
    ...  
  ENDLAYOUT ;  
END ;  
RUN ;
```

1.4 - Syntaxe (2/2)

```
ODS PATH bib.itemStore
ODS HTML|PDF|RTF ... ;
DATA _NULL_ ;
    SET tableSAS ;
    FILE PRINT ODS=(TEMPLATE="nomGraphique"
        < DYNAMIC=(
            nomVar1="valeur1"
            < nomVar2="valeur2" >
            < ... > > )) ;
    PUT _ODS_ ;
RUN ;
```

2- Éléments graphiques



© Olivier Decourt 2008
516 éditions Dunod
Comptabilité à Reporting avec

2.1 - Éléments graphiques (1/2)

Type d'élément	Sur données brutes		Sur statistiques	
	Base	A superposer	Base	A superposer
Nuage de points	SCATTERPLOT + NEEDLEPLOT + SCATTERPLOTMATRIX	SCATTER + NEEDLE		
Courbe	SERIESPLOT + STEPLOT + DENSITYPLOT	SERIES + STEP + DENSITY		LINEPARM (droite)
Histogramme	HISTOGRAM	HISTO	BARCHARTPARM + HISTOGRAMPARM	BARPARM + HISTOPARM
Boxplot	BOXPLOT	BOX	BOXPLOTPARM	BOXPARM
Intervalle de confiance	BANDPLOT	BAND + ELLIPSE		ELLIPSEPARM
Graphique rayonnant (plan factoriel des variables en ACP)	VECTORPLOT	VECTOR		

2.1 - Éléments graphiques (2/2)

- Élément de base :
 - indispensable dans tous les motifs
 - un et un seul par motif
- Élément à superposer :
 - uniquement dans un motif LAYOUT OVERLAY
 - autant qu'on veut dans un même motif
- Définition des axes :
 - dans LAYOUT si c'est OVERLAY
 - dans l'élément de base pour un autre LAYOUT

2.2 - Gestion des axes (1/2)

... / XAXISOPTS | YAXISOPTS=(options) ... ;

- DISPLAY=ALL|NONE|(élément(s)) où les éléments à afficher sont LABEL, LINE (=l'axe lui-même), TICKS (=les graduations) et VALUES
- TICKS=(valeurs) pour forcer les valeurs des graduations
- LABEL="intitulé" pour forcer le libellé de l'axe. En revanche, LABEL=NONE ne fonctionne pas ; il faut écrire LABEL=" "

2.2 - Gestion des axes (2/2)

XGRID, YGRID=TRUE FALSE	Lignes de références verticales, horizontales
YEQUATED=TRUE FALSE	Graphique avec des échelles identiques en X et Y
XMIN, XMAX, YMIN, YMAX = valeur	Amplitude des axes
XAXIS=X X2, YAXIS=Y Y2	Axe affiché en bas/en haut, à gauche/à droite

2.3 - L'élément BOXPLOT (1/3)

```
BOXPLOT Y=variableEtudiee  
  < X=variableGroupeBoxplots > /  
  DATA LABEL=variableID  
  LABEL FAR=TRUE | FALSE  
  ORIENT=HORIZONTAL | VERTICAL ;
```

Marque les points qui s'éloignent du centre de la distribution


Orientation de la boxplot

Variable contenant les libellés identifiant les points

2.3 - L'élément BOXPLOT (2/3)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.boxplot ;
    DYNAMIC varY varGpe ;
    LAYOUT GRIDDED ;
      BOXPLOT Y=varY X=varGpe ;
    ENDLAYOUT ;
  END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale ;
  FILE PRINT ODS=(TEMPLATE="exemples.boxplot"
                  DYNAMIC=(varY="actual"
                           varGpe="country" )) ;
  PUT _ODS_ ;
RUN ;
```

2.3 - L'élément BOXPLOT (3/3)



2.4 - L'élément HISTOGRAM (1/5)

HISTOGRAM variableEtudiee /

{
 BINSTART=valeur
 BINWIDTH=valeur
 NBINS=nbBarres
 FREQ=varPoids
 ORIENT=HORIZONTAL | VERTICAL
 SCALE=COUNT | PERCENT | PROPORTION ;

Pondération

Définition des intervalles
associés aux barres :
minimum, largeur, nombre


Type de graduation
de l'axe vertical

Complément

2.4 - L'élément HISTOGRAM (2/5)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.histo ;
    DYNAMIC varY ;
    LAYOUT GRIDDED ;
      HISTOGRAM varY / SCALE=PERCENT BINWIDTH=200 ;
    ENDLAYOUT ;
  END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale ;
  FILE PRINT ODS=(TEMPLATE="exemples.histo"
                  DYNAMIC=(varY="actual")) ;
  PUT _ODS_ ;
RUN ;
```


2.4 - L'élément HISTOGRAM (3/5)



2.4 - L'élément HISTOGRAM (4/5)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.histo ;
    DYNAMIC varY ;
    LAYOUT GRIDDED ;
      HISTOGRAM varY / SCALE=PERCENT BINWIDTH=200
                          XAXISOPTS=(LABEL=varY) ;
    ENDLAYOUT ;
  END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale ;
  FILE PRINT ODS=(TEMPLATE="exemples.histo"
                  DYNAMIC=(varY="actual")) ;
  PUT _ODS_ ;
RUN ;
```

2.4 - L'élément HISTOGRAM (5/5)



2.5 - L'élément DENSITYPLOT (1/3)

```
DENSITYPLOT variableEtudiee /  
  KERNEL ( ) | NORMAL ( ) ;
```


Densité empirique

Densité d'une loi normale de
mêmes moyenne et écart-
type

2.5 - L'élément DENSITYPLOT (2/3)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.histoDensite ;
  DYNAMIC varY ;
  LAYOUT OVERLAY / XAXISOPTS=(LABEL=varY) ;
  HISTOGRAM varY / SCALE=PERCENT BINWIDTH=200 ;
  DENSITY varY / NORMAL() ;
  ENDLAYOUT ;
END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale ;
  FILE PRINT ODS=(TEMPLATE="exemples.histoDensite"
                  DYNAMIC=(varY="actual")) ;
  PUT _ODS_ ;
RUN ;
```

2.5 - L'élément DENSITYPLOT (3/3)



Compl

2.6 - L'élément SCATTERPLOT (1/4)

SCATTERPLOT Y=variableOrdonnees

X=variableAbscisses /

DATALABEL=variableID

















MARKERSYMBOL=symbole | CHARACTER

GROUP=variableBy ;

Une série par
valeur de cette
variable

Symbole utilisé
pour marquer
l'emplacement des
points


2.6 - L'élément SCATTERPLOT (2/4)

Nom	Symbole	Nom	Symbole
CIRCLE		FEMALEFIGUREFILLED	
CIRCLEFILLED		MALEFIGUREFILLED	
CROSS		MARSBOLD	
DIAMOND		VENUSBOLD	
DIAMONDNARROW		STAR	
FLAG		TRIANGLE	
PLUS		TRIANGLEDOWN	
SALTIRE		SQUAREX	

2.6 - L'élément SCATTERPLOT (3/4)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.nuage ;
  DYNAMIC varX varY symbole ;
  LAYOUT GRIDDED ;
    SCATTERPLOT X=varX Y=varY /
      MARKERSYMBOL=symbole
      MARKERSIZE=20 ;
  ENDLAYOUT ;
END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.class (WHERE=(sex="F")) ;
  FILE PRINT ODS=(TEMPLATE="exemples.nuage"
    DYNAMIC=(varX="weight"
      varY="height" symbole="FEMALEFIGUREFILLED")) ;
  PUT _ODS_ ;
RUN ;
```

2.6 - L'élément SCATTERPLOT (4/4)



2.7 - L'élément SCATTERPLOTMATRIX (1/3)

```
SCATTERPLOTMATRIX variable1 variable2  
                  < variable 3 < ... > > /
```

```
DATA LABEL=variableID
```

```
GROUP=variableBy
```

```
MARKER SYMBOL=symbole | CHARACTER ;
```

Symbole utilisé pour marquer l'emplacement des points

Une série par valeur de cette variable


Variable contenant les libellés identifiant les points

2.3 - L'élément SCATTERPLOTMATRIX (2/3)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.nuagesMultiples ;
    DYNAMIC v1 v2 v3 v4 v5 ;
    LAYOUT GRIDDED ;
      SCATTERPLOTMATRIX v1 v2 v3 v4 v5 ;
    ENDLAYOUT ;
  END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.shoes ;
  FILE PRINT ODS=(TEMPLATE="exemples.nuagesMultiples"
    DYNAMIC=(v1="inventory"
      v2="sales"
      v3="returns")) ;
  PUT _ODS_ ;
RUN ;
```

2.3 - L'élément SCATTERPLOTMATRIX

(3/3)



Complé

2.8 - L'élément SERIESPLOT (1/5)

SERIESPLOT Y=variableOrdonnees

X=variableAbscisses /

DATA LABEL=variableID

GROUP=variableBy

MARKER SYMBOL=symbole | CHARACTER

LINE COLOR=couleur

LINE THICKNESS=dimension

LINE PATTERN=SOLID | DASH | DOT ;

Symbole utilisé pour marquer l'emplacement des points

Variable contenant les libellés identifiant les points


Aspect de la courbe

Une série par valeur de cette variable

2.8 - L'élément SERIESPLOT (2/5)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.courbe ;
  DYNAMIC varX varY varBy ;
  LAYOUT GRIDDED ;
    SERIESPLOT X=varX Y=varY / GROUP=varBy ;
  ENDLAYOUT ;
END ;
RUN ;
ODS OUTPUT summary=work.ventes ;
PROC MEANS DATA=sashelp.prdsale MEAN ;
  VAR actual ;
  CLASS month country ;
RUN ;
DATA _NULL_ ;
  SET work.ventes ;
  FILE PRINT ODS=(TEMPLATE="exemples.courbe"
                  DYNAMIC=(varX="month" varY="actual_mean"
                            varBy="country")) ;
  PUT _ODS_ ;
RUN ;
```

2.8 - L'élément SERIESPLOT (3/3)



2.9 - L'élément LINEPARM (1/6)

YINTERCEPT seul → droite horizontale (VREF)
XINTERCEPT seul → droite verticale (HREF)
YINTERCEPT + SLOPE → droite oblique
(ordonnée à l'origine + pente)

LINEPARM YINTERCEPT | XINTERCEPT=valeur

SLOPE=valeur /

LINECOLOR=couleur ←

→ LINETHICKNESS=dimension

LINEPATTERN=SOLID | DASH | DOT ;

Couleur de la droite

Epaisseur de la droite
(2%, 2pt, 2px, ...)

Type de trait : plein,
discontinu, pointillé


2.9 - L'élément LINEPARM (2/6)

```
ODS OUTPUT parameterEstimates = work.coeffs ;
PROC REG DATA = sashelp.prdsale ;
  WHERE country = "CANADA"
    AND product="CHAIR"
    AND region = "WEST"
    AND division="CONSUMER"
    AND year=1993 ;
  MODEL actual = month ;
RUN ; QUIT ;
DATA _NULL_ ;
  SET work.coeffs ;
  IF variable="Intercept" THEN mv = "constante" ;
  ELSE mv = "coeffX" ;
  CALL SYMPUTX(mv, estimate) ;
RUN ;
```

2.9 - L'élément LINEPARM (3/6)

```
PROC TEMPLATE ;
  DEFINE statGraph exemples.nuageTendance ;
    DYNAMIC varX varY ;
    MVAR constante coeffX ;
    LAYOUT OVERLAY ;
      SCATTERPLOT X=varX Y=varY / MARKERSYMBOL=STAR ;
      LINEPARM YINTERCEPT=constante SLOPE=coeffX ;
    ENDLAYOUT ;
  END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale ;
  WHERE country="CANADA" AND product="CHAIR"
    AND region="WEST" AND division="CONSUMER"
    AND year=1993 ;
  FILE PRINT ODS=(TEMPLATE="exemples.nuageTendance"
    DYNAMIC=(varX="month" varY="actual")) ;
  PUT _ODS_ ;
RUN ;
```


2.9 - L'élément LINEPARM (4/6)



2.9 - L'élément LINEPARM (5/6)

```
PROC TEMPLATE ;
  DEFINE statGraph exemples.nuageTendanceSansReg ;
    DYNAMIC varX varY ;
    LAYOUT OVERLAY ;
      SCATTERPLOT X=varX Y=varY / MARKERSYMBOL=STAR ;
      LINEPARM SLOPE=EVAL((MEAN(varX*varY)-MEAN(varX)*MEAN(varY))/VAR(varX))
        YINTERCEPT=EVAL(MEAN(varY)-MEAN(varX)*
          (MEAN(varX*varY)-MEAN(varX)*MEAN(varY))/VAR(varX)) ;
    ENDLAYOUT ;
  END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale ;
  WHERE country="CANADA" AND product="CHAIR"
    AND region="WEST" AND division="CONSUMER"
    AND year=1993 ;
  FILE PRINT ODS=(TEMPLATE="exemples.nuageTendanceSansReg"
    DYNAMIC=(varX="month" varY="actual")) ;
  PUT _ODS_ ;
RUN ;
```

2.9 - L'élément LINEPARM (6/6)





3- Éléments de texte

© Olivier Decollet 2008
à Reporting avec 5/5 éditions Dunod

3.1 - Éléments graphiques

- **ENTRY et ENTRYTITLE**
 - insertion de lignes de texte (**ENTRYTITLE** : mis en gras)
- **DISCRETELEGEND**
 - légende permettant de repérer des éléments juxtaposés ou contigus
 - personnalisable
 - associée à des noms donnés aux éléments de la légende : option **NAME** dans les éléments graphiques

3.2 - Les éléments ENTRY et ENTRYTITLE (1/3)

```
ENTRY | ENTRYTITLE "texte" /  
    { VALIGN=TOP | CENTER | BOTTOM  
      HALIGN=LEFT | CENTER | RIGHT  
      PADTOP=marge  
      PADBOTTOM=marge  
      PADLEFT=marge  
      PADRIGHT=marge } ;
```


Emplacement
(par défaut,
au centre !)

Marges autour du texte :
en pixels (2px), en points
(2pt), en pourcentage de
l'aire graphique (2%)

3.2 - Les éléments ENTRY et ENTRYTITLE (2/3)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.histo ;
    DYNAMIC varY ;
    LAYOUT GRIDDED ;
      ENTRYTITLE "Répartition de la variable" ;
      HISTOGRAM varY / SCALE=PERCENT BINWIDTH=200
        XAXISOPTS=(LABEL=varY) ;
    ENDLAYOUT ;
  END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale ;
  FILE PRINT ODS=(TEMPLATE="exemples.histo"
    DYNAMIC=(varY="actual")) ;
  PUT _ODS_ ;
RUN ;
```

3.2 - Les éléments ENTRY et ENTRYTITLE (3/3)



Compléter

3.3 - L'élément DISCRETELEGEND (1/5)

DISCRETELEGEND noms /

ACROSS=nombreColonnes

DOWN=nombreLignes

Présence
d'une bordure

BORDER=TRUE|FALSE

Disposition
des éléments
de la légende

Texte en tête
de légende

TRANSPARENCY=nombre

TITLE=texte

HALIGN=LEFT|CENTER|RIGHT

VALIGN=TOP|CENTER|BOTTOM ;


Emplacement
(par défaut,
au centre !)

Transparence de la légende
de 0 (opaque) à 1 (transparent)

3.3 - L'élément DISCRETELEGEND (2/5)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.courbe ;
  DYNAMIC varX varY varBy ;
  LAYOUT GRIDDED ;
    SERIESPLOT X=varX Y=varY / GROUP=varBy NAME="courbes" ;
    DISCRETELEGEND "courbes" / BORDER=TRUE TITLE=varBy ;
  ENDLAYOUT ;
END ;
RUN ;
ODS OUTPUT summary=work.ventes ;
PROC MEANS DATA=sashelp.prdsale MEAN ;
  VAR actual ;
  CLASS month country ;
RUN ;
DATA _NULL_ ;
  SET work.ventes ;
  FILE PRINT ODS=(TEMPLATE="exemples.courbe"
                  DYNAMIC=(varX="month" varY="actual_mean"
                            varBy="country")) ;
  PUT _ODS_ ;
RUN ;
```

3.3 - L'élément DISCRETELEGEND (3/5)




Compl

3.3 - L'élément DISCRETELEGEND (4/5)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.distrib ;
  DYNAMIC varX ;
  LAYOUT OVERLAY ;
    HISTOGRAM varX ;
    DENSITY varX / NORMAL() LINEPATTERN=DASH
      NAME="norm" LEGENDLABEL="Loi normale" ;
    DENSITY varX / KERNEL() LINEPATTERN=SOLID
      NAME="empirique" LEGENDLABEL="Loi empirique" ;
    DISCRETELEGEND "norm" "empirique" /
      BORDER=TRUE ACROSS=1 VALIGN=TOP HALIGN=RIGHT ;
  ENDLAYOUT ;
END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale;
  FILE PRINT ODS=(TEMPLATE="exemples.distrib"
    DYNAMIC=(varX="actual")) ;
  PUT _ODS_ ;
RUN ;
```

3.3 - L'élément DISCRETELEGEND (5/5)





Compl

4- Juxtaposition d'éléments



© Olivier Decourt 2008
Reporting avec 5/6 éditions Dunod



4.1 - Le motif LAYOUT LATTICE (1/3)


LAYOUT GRIDDED

```
< / ROWS=nbLignes  
  COLUMNS=nbColonnes > ;
```

...

```
ENDLAYOUT ;
```


Définition du
quadrillage



4.1 - Le motif LAYOUT LATTICE (2/3)

```
PROC TEMPLATE ;
  DEFINE STATGRAPH exemples.distrib ;
  DYNAMIC varX ;
  LAYOUT OVERLAY ;
    HISTOGRAM varX ;
    DENSITY varX / NORMAL() LINEPATTERN=DASH
      NAME="norm" LEGENDLABEL="Loi normale" ;
    DENSITY varX / KERNEL() LINEPATTERN=SOLID
      NAME="empirique" LEGENDLABEL="Loi empirique" ;
    DISCRETELEGEND "norm" "empirique" /
      BORDER=TRUE ACROSS=1 VALIGN=TOP HALIGN=RIGHT ;
  ENDLAYOUT ;
END ;
RUN ;
DATA _NULL_ ;
  SET sashelp.prdsale;
  FILE PRINT ODS=(TEMPLATE="exemples.distrib"
    DYNAMIC=(varX="actual")) ;
  PUT _ODS_ ;
RUN ;
```

4.1 - Le motif LAYOUT LATTICE (3/3)



Compl

4.2 - Le motif LAYOUT GRIDDED (1/6)

LAYOUT GRIDDED


/ ROWS=nbLignes

COLUMNS=nbColonnes

Définition du quadrillage

ROWSPAN | COLSPAN = (...)


...
ENDLAYOUT ;



4.2 - Le motif LAYOUT GRIDDED (2/6)

```
PROC TEMPLATE ;
  DEFINE statGraph exemples.courbesComparees ;
    DYNAMIC varY1 varY2 varX ;
    LAYOUT GRIDDED / COLUMNS=2 ROWS=2 COLUMNSPAN=(1 1 2) ;
      SERIESPLOT Y=varY1 X=varX ;
      SERIESPLOT Y=varY2 X=varX / YAXIS=Y2 ;
      LAYOUT OVERLAY / YAXISOPTS=(LABEL="Ventes") ;
        SERIESPLOT Y=varY1 X=varX / LINEPATTERN=SOLID
          LEGENDLABEL=varY1
          NAME="y1" ;
        SERIES Y=varY2 X=varX / LINEPATTERN=DASH
          LEGENDLABEL=varY2
          NAME="y2" ;
        DISCRETELEGEND "y1" "y2" / VALIGN=BOTTOM HALIGN=LEFT ;
      ENDLAYOUT ;
    ENDLAYOUT ;
  END ;
RUN ;
```

4.2 - Le motif LAYOUT GRIDDED (3/6)



Compl


4.2 - Le motif LAYOUT GRIDDED (4/6)

```
PROC TEMPLATE ;
  DEFINE statGraph exemples.courbesComparees ;
  DYNAMIC varY1 varY2 varX ;
  LAYOUT GRIDDED / COLUMNS=2 ROWS=2 COLUMNSPAN=(1 1 2) ;
  LAYOUT OVERLAY ;
    SERIESPLOT Y=varY1 X=varX ;
    LINEPARM SLOPE=EVAL((MEAN(varX*varY1)-MEAN(varX)*MEAN(varY1))/VAR(varX))
    YINTERCEPT=EVAL(MEAN(varY1)-MEAN(varX)*
      (MEAN(varX*varY1)-MEAN(varX)*MEAN(varY1))/VAR(varX))
    ;
  ENDLAYOUT ;
  LAYOUT OVERLAY ;
    SERIESPLOT Y=varY2 X=varX ;
    LINEPARM SLOPE=EVAL((MEAN(varX*varY2)-MEAN(varX)*MEAN(varY2))/VAR(varX))
    YINTERCEPT=EVAL(MEAN(varY2)-MEAN(varX)*
      (MEAN(varX*varY2)-MEAN(varX)*MEAN(varY2))/VAR(varX))
    ;
  ENDLAYOUT ;
```

4.2 - Le motif LAYOUT GRIDDED (5/6)

```
LAYOUT OVERLAY / YAXISOPTS=(LABEL="Ventes") ;  
  SERIESPLOT Y=varY1 X=varX / LINEPATTERN=SOLID  
    LEGENDLABEL=varY1  
    NAME="y1" ;  
  SERIES      Y=varY2 X=varX / LINEPATTERN=DASH  
    LEGENDLABEL=varY2  
    NAME="y2" ;  
  DISCRETELEGEND "y1" "y2" / VALIGN=BOTTOM HALIGN=LEFT ;  
ENDLAYOUT ;  
ENDLAYOUT ;  
END ;  
RUN ;
```

4.2 - Le motif LAYOUT GRIDDED (6/6)



Compl